/*	This program produces Appendix Figure C.2 */

***** Set directories 
local dir_clean 	"~/Dropbox/Retirement gaming/clean"
local dir_output 	"~/Dropbox/Retirement gaming/output/dataverse"


use "`dir_clean'/mainsample_medbcw.dta", clear

global listcontrols = " i.t i.ciiu2_1stobs i.ndep_cat_1stobs i.ndep_cat i.year#i.i.ciiu2_1stobs i.year#i.ndep_cat_1stobs "  

** SELF EMPLOYED REPORTS MCB (Figure C.2) **
preserve
replace time_bcw=-5 if time_bcw==-6
keep if self_empl==1 
* Age dummies
tab time_bcw, gen(time_bcw_dums)
drop time_bcw_dums5 // drop year before start BCW
* Regression
reghdfe reports_ficto time_bcw_dums*, absorb(${listcontrols}) vce(cluster i)
* Sample size
global n=e(N)
global N=e(N_clust1)
* Coefficients and sd
for any beta sd: gen X = .
forvalues d = 1(1)13 {
	capture qui replace beta = _b[time_bcw_dums`d'] if time_bcw==`d'-6
	capture qui replace sd   = _se[time_bcw_dums`d'] if time_bcw==`d'-6
}
replace beta=0 if time_bcw==-1
replace sd=0 if time_bcw==-1
* Plot the results
collapse beta* sd*, by(time_bcw)
gen sd_top = beta + 1.96*sd	
gen sd_bot = beta - 1.96*sd
gen sd_top10 = beta + 1.645*sd	
gen sd_bot10 = beta - 1.645*sd
twoway rcap sd_top sd_bot time_bcw, ///
	cmissing(n) lwidth(thin) lcolor(navy) xsc(r(-5 7)) xlabel(-5(1)7)  xline(-1, lc(green)) ysc(r(-0.04 .04)) ylabel(-0.04(0.02).04,grid)  ///
	|| rcap sd_top10 sd_bot10 time_bcw, cmissing(n) lwidth(thin) lcolor(navy) xsc(r(-5 7)) ///
	|| scatter beta time_bcw, mcolor(navy) scheme(s1color) lcolor(navy) lpattern(solid) yline(0, lcolor(gs2)) ///
	title("Probability of reporting Min. Contribution Base") note("N=${n}, Individuals=${N}") ///
	legend(off) ytitle("Estimated Coefficients") xtitle("Years Relative to Predicted Start of BCW")   
graph export "`dir_output'/figureC2.png",  replace 	
restore		

 


clear all
exit
 
